

数式処理ソフトMaple入門(III)

一式と変形一

関西学院大学理工学部 西谷滋人

Copyright ©2009 by Shigeto R. Nishitani

等号の意味を問う

数式処理ソフトを習得するのは、プログラミング言語を習得するのとは少し違ったセンスが必要です。しかしそのセンスはなにもむずかしい構文を新たに覚えるのではなく、これまで積み上げてきた数学の知識にほんの少し修正を加えるだけです。この修正は中学の1, 2年生あたりで導入される代数計算、あるいは式変形の約束と似たところがあります。

中学あたりでつまずく学生の多くが、式の変形、代入、方程式を混同しているようです。例えば、よくある中学生の間違いとして、

$$3x - x = 3$$

ってのがあります。これは $3x$ が $3 \times x$ の略記であることを知らずに「3とxからxを引いたら3が残る」と思っているそうです。たしかに小学校の

$$3\frac{1}{2} - \frac{1}{2} = 3$$

なら正しい計算です。期待している正しい「変形」は

$$3x - x = 2x$$

です。ところが、元の式も方程式と見なせば正しく、その辺りの混乱を引きずって知識を積み上げているようです。

もうひとつ、プログラミングを大学で教えていて、なかなか納得しない学生のなかに、

$$x = x + 1$$

は明らかに「式の変形および方程式」として間違っているからというのがいてます。刷り込まれた知識を否定するのはことのほか困難なようです。

でも、いざ教えるとなると私も「ん～～、プログラミング言語の仕様上そうになっているから自分で適当に納得してね。」となります。これら諸悪の根源は、等号「=」を無節操に使い過ぎるせいかもしれません。コンピュータに解釈させなければいけない数式処理ソフトMapleではこのような曖昧さが許されません。厳密で融通がききませんが、かえって区別がつきやすいかも。そのあたりからMapleのコマンドの癖を見ていきましょう。

等号1：代入

たとえば、

$a=3, b=2$ のとき、 $3a+4b$ の式の値はいくらか。

という問題をMapleに教えてやる場合には、

```
> a:=3;  
b:=2;  
3*a+4*b;
```

$a := 3$

$b := 2$

17

(3.1)

となります。等号の代わりに使われている記号“:=”は、代入を意味します。こういう解釈で問題を読み直すと、

a に3を、 b に2を代入したとき、 $3a+4b$ の式の値はいくらか。

とすべきです。Mapleではかけ算も含めて省略することは許されないし、等号の意味、ここでは変数 a, b に数値を代入すべしとすることを明示的に教える必要があります。

これに、間違っても等号を使ったとしても

```
> restart;  
a=3;  
b=2;  
3*a+4*b;
```

$a = 3$

$b = 2$

$3a + 4b$

(3.2)

となり、なにも意味のある答えを返してきません。 $a=3$ や $b=2$ は単にこういう式があると解釈され、 a, b に数値が代入されることはありません。

ちなみに、`restart`は上の方で a, b に代入した値を忘れさせるための初期化(`restart`)操作であり、行末の“;”はMapleが解釈すべきひとかたまりの入力の区切りを意味しています。改行だけするのは、`shift+enter`(`return`)です。赤い箇所のどこにカーソルがあってもそこで`enter`(`return`)を入れると、そのブロック全体が入力、Mapleに反応を返すように促します。それ以外の操作は単にエディターで編集しているだけです。

Mapleというソフトは編集画面付きのインタープリターだと思ってください。C言語では、エディターでプログラムを作って保存してコンパイルして実行して結果が返ってきますが、Mapleではリターンキーでそれらの操作が一瞬で終わるわけです。Mapleのメモリーは`return`で入力された順にたまっていき、画面に出ている順番とは本来関係ありません。また、`restart`でメモリークリアされます。

▼ 等号 2 : 方程式を解く

方程式の場合でも同様に、

$3x-x=3$ の方程式を満たす x はいくらか、

は

$3*x-x=3$ を x の方程式として解け(solve).

と読み直して、

```
> solve(3*x-x=3,x);
```

$$\frac{3}{2}$$

(4.1)

となります。Mapleのコマンドはすべてこのような形をしています。つまり、一般形で書くと

```
command(引数1, 引数2, . . . );
```

あるいは

```
command(引数1, オプション1, オプション2, . . . );
```

となります。C言語などの手続き型プログラミング言語の標準的なフォーマットと同じです。丸括弧、コンマ、セミコロンなどの区切りとしての意味を意識してください。

もうすこし手順がわかるように分解して、

まず変数eq1に式 $3*x-x=3$ を代入し、

次にeq1を x について解け。

と考えます。これをMapleに教えると、

```
> eq1:=3*x-x=3;
```

```
solve(eq1,x);
```

$$eq1 := 2 x = 3$$

$$\frac{3}{2}$$

(4.2)

英語では、

```
solve eq1 for x.
```

となるので、Mapleのコマンド順序そのままなのですが...

▼ 等号 3 : 式の変形

式の変形にも等号が使われます。式の変形は、どんな x についても成り立つ恒等式を意図しています。例えば、

$$(x-2)^2 = x^2 - 4x + 4$$

というのが等号で結ばれています。

この式変形も、Mapleでは明示的に

$(x-2)^2$ を展開(expand)せよ。

と読み直して、

```
> expand((x-2)^2);
```

$$x^2 - 4x + 4$$

(5.1)

とします。変形した結果を残すには、それを変数に代入します。つまり、

```
> eq2:=expand((x-2)^2);
```

```
eq2;
```

$$eq2 := x^2 - 4x + 4$$

$$x^2 - 4x + 4$$

(5.2)

因数分解は

$x^2 - 3x + 2$ を因数分解(factor)せよ。

```
> factor(x^2-3*x+2);
```

$$(x-1)(x-2)$$

(5.3)

微分や積分も等号で結ばれますが、微分(differentiate)せよあるいは積分(integrate)せよという命令を明示します。

$$\frac{d}{dx}(2x-3) = 2$$

あるいは

$$\int 2x-3 dx = x^2 - 3x$$

というのは、微分・積分変数 x を第2引数として、

```
> diff(2*x-3,x);
```

```
int(2*x-3,x);
```

$$2$$

$$x^2 - 3x$$

(5.4)

さらに定積分

$$\int_1^3 2x-3 dx$$

は、第2引数をいじって、

```
> int(2*x-3,x=1..3);
```

$$2$$

(5.5)

ここでMapleコマンドの新しい表記法が出てきました。これは、どこからどこまでの範囲を表す、

$$x=1..3$$

".."です。

数学の問題では

$1/a+1/b$ を簡単にせよ。

というとても他人任せの問題があります。一応Mapleにも

```
> simplify(1/a+1/b);
```

$$\frac{b+a}{ab} \quad (5.6)$$

というコマンドが用意されています。しかし、式変形あるいは簡単にするという操作は、主観や状況によって変わってきます。どうするかを明示的にMapleに教えてやってください。

上の変形は、通分(normal)操作で

> normal(1/a+1/b);

$$\frac{b+a}{ab} \quad (5.7)$$

とかけます。

ax+bx-cを同類項でまとめよ。

というちょっと曖昧な問題は

a*x+b*x-cをxの同類項でまとめよ(collect).

と読み直して、

> collect(a*x+b*x-c,x);

$$(b+a)x-c \quad (5.8)$$

係数(coefficient)を取り出すコマンドもあります。この場合、引数は3つ取れます。最後の"1"はxの1次項という意味です。このあたりを使うと恒等式とか式の変形とかが柔軟に扱えます。

> coeff(a*x+b*x-c,x,1);

$$a+b \quad (5.9)$$

▼ 等号4：関数定義

等号のもう一つの重要な使い方に関数の定義があります。

関数 $f(x)$ を $f(x) = 2x - 3$ とおく

とするやつです。Mapleでの関数定義は、

> f:=x->2*x-3;

$$f:=x \rightarrow 2x-3 \quad (6.1)$$

と、矢印で書く必要があります。これが関数として定義されているかどうかを確認するには、いくつかの数値や変数を $f(x)$ に代入してみれば確認できます。


> f(3);

f(a);

$$\begin{aligned} & 3 \\ & 2a-3 \end{aligned} \quad (6.2)$$

うまく行っているようです。さらに $y = f(x)$ と考えてプロットさせてみましょう。

> plot(f(x),x=-2..2);



ここでも、どこからどこまでの範囲を表す、

x=-2..2

を使っています。

もう一つ関数定義のコマンドとして次のもの同じ意味ですので、覚えておいてください。

> f:=unapply(2*x-3,x);

$$f:=x \rightarrow 2x-3 \quad (6.3)$$

▼ Mapleでの等号まとめ

単純に言っちゃえば、Mapleでは等号、代入、関数定義を"=",":=","->"で区別することになります。他に、Mapleコマンドによるプログラムで使う関係演算子がありますが、これはまた後で。3回のsuper introductionでMapleの基本的操作の説明は終わりです。これよりややこしいのは、数学のせいであって、数式処理ソフトのせいではありません。そのあたりを誤解しないように、Mapleの操作法がわからないのか、数学がわからないのかの違いを常に意識してください。ややこしいコマンドや関数はヘルプを参照してください。